AN UNIVERSAL MODEL FOR LOSSY AND DISPERSIVE TRANSMISSION LINES FOR TIME DOMAIN CAD OF CIRCUITS

José I. Alonso, José Borja and Félix Pérez

Departamento de Señales, Sistemas y Radiocomunicaciones
ETSI. Telecomunicación. Ciudad Universitaria. 28040 Madrid. SPAIN.

ABSTRACT

An universal equivalent circuit for lossy and dispersive transmission lines is presented. Existing CAD packages, such as SPICE, can be used for its implementation. The starting point for obtaining the model are the analog filters which approximate the forward impulse response and characteristic impedance. The equivalent circuit is used to simulate the effects for pulse propagation on microstrip transmission lines. An examination of the validity of the model is carried out analyzing the response for an example case in the time and frequency domains.

INTRODUCTION

As circuit speeds increase, more attention must be paid to the design of interconnects. Improperly designed interconnects can result in increases of signal delay because of losses, inadvertent switching and noise as a result of crosstalk [1]. With subnanosecond rise times, the length of interconnects can become a significant fraction of a wavelength. Consequently, the conventional microstrip transmission line models, which generally assume ideal lines, are not adequate in this case. In [2] a model, compatible with SPICE, for lossy and dispersive transmission lines has been proposed. This model, however, requires a great quantity of lumped elements sections and therefore is not useful in most practical cases.

In this paper a technique which allows to obtain an universal equivalent circuit in order to model this class of structures is presented.

TRANSMISSION-LINE MODELLING TECHNIQUE

Basic principles

Lossy and dispersive transmission lines have parameters $-Z_o(s), \gamma(s)-$ dependent on frequency. In order to develop a general circuit model for time domain CAD, it is necessary to simulate these effects efficiently. This involves synthesizing a symmetrical network with both, a frequency dependent characteristic impedance and a lossy and nonlinear phase propagation function. Standard circuit design synthesis methods foreither frequency dependent one-ports (impedances) or two-ports with any propagation function but with a constant characteristic impedance exist [3]. As it will be shown, in the proposed technique the general problem is broke into these two using an adequate network that includes linear voltage dependent voltage sources and the so designed impedances and two-ports. An efficient way of designing those impedances and two-ports with few lumped elements and ideal transmission lines is also developed.

Implementation of the transmissión line model

a) <u>Generalized method of characteristics</u>

Transmission lines can be characterized by the following equations:

$$V_1(s) = Z_o(s)I_1(s)+F(s)[V_2(s)+Z_o(s).I_2(s)] \quad (1)$$
$$V_2(s) = Z_o(s)I_2(s)+F(s)[V_1(s)+Z_o(s).I_1(s)] \quad (2)$$

where $F(s)=\exp[-\gamma(s)l]$ is the propagation function and $(V_1(s),I_1(s))$ and $(V_2(s),I_2(s))$ are the terminal voltages and currents at the near and far end of the transmission line, as shown in fig.1.



Fig.1. Transmission line relationships

The previous equations suggest using a transmission line model as shown in fig.2.



Fig.2. The characteristic model of a transmission line

where:

$$V_a(s) = Z_o(s).I_1(s) \quad (3)$$
$$V_b(s) = Z_o(s).I_2(s) \quad (4)$$
$$E_1(s) = F(s).V_P(s) = F(s)[V_2(s)+V_b(s)] \quad (5)$$
$$E_2(s) = F(s).V_Q(s) = F(s)[V_1(s)+V_a(s)] \quad (6)$$

Networks that implement the impedance -$Z_o(s)$- and the voltage sources -$E_1(s)$,$E_2(s)$- must be found.

The first is an impedance synthesis problem. The voltage $E_1(s)$ -or $E_2(s)$- can be obtained from the propagation of the voltage $2V_P(s)$ -or $2V_Q(s)$- accross a two-port as long as its propagation function is F(s) and its arbitrary characteristic impedance, R_n, is the same as the circuit loads, as shown in fig.3.



Fig.3. Circuits to obtain the voltages E_1 and E_2

If in the circuit of fig.2 we replace the sources E_1 and E_2 by linear unity gain voltage controlled voltage sources depending upon the E_1 and E_2 voltages of fig.3 we will have solved our problem as the impedance and the two-port can be designed with standard methods. However, as circuits 3a and 3b are the same except for the excitation, the superposition principle can be used to obtain the results shown in fig.4.

$V_{ta} = E_1 - V_Q$ $\qquad$ $V_{tb} = E_2 - V_P$



Fig.4. Application of superposition principle on circuit of previous figure

Now V_{ta} and V_{tb} are obtained instead of E_1 and E_2, and therefore, aditional voltage sources must be placed in the circuit of fig.2 adding V_P and V_Q to V_{ta} and V_{tb}, giving finally the circuit of fig.5.

b) <u>Synthesis of the Characteristic Impedance and the Propagation Function</u>

In accordance with the method described in [4], the analog filters used for $F(s)=\exp[-\gamma(s)l]$ and $Z_o(s)$ that take account of dispersive effects of the modeled transmission line, are:

$$F(s) = k_1 \prod_{j=1}^{N_1} \frac{(s+c_j)}{(s+d_j)} \prod_{i=1}^{N_2} \frac{(s^2-f_{1i}s+f_{0i})}{(s^2+f_{1i}s+f_{0i})} \exp(-s\tau) \quad (7)$$

and

$$Z_o(s) = k_2 \prod_{j=1}^{N_2} \frac{s^2+g_j s+h_j}{s^2+e_j s+b_j} \quad (8)$$

The term $\exp(-s\tau)$ takes into account a constant delay, implemented with an ideal transmission line, so that the rest of the frequency dependent propagation delay is easier to obtain with simpler networks.

The coefficients of the previous expressions are found by a least-squares matching procedure from available data in the frequency domain for the effective dielectric constant [5] and characteristic impedance, or starting from values of R, L G and C of the line [6]. Stability and realizability considerations must be taken into account on the minimisation procedure.

The lumped elements of the model are obtained by synthesis of functions (7) and (8) by conventional techniques.

COMPUTATIONS

In order to validate the accuracy of the modeling technique and demonstrate the usefulness of the model, the distorsion of a 45 psg wide -at the half magnitude points- gaussian pulse travelling along a 122.142mm long microstrip line is computed. The line substrate is alumina (ε_r=10.2), and the line parameters are: H=0.635mm, t=0.0017mm, $R_s=8.63\times10^{-3}\sqrt{f}$ and $\tan\delta$=0.0015.

Figure 6 shows good agreement between the propagation characteristics of the obtained circuit and values computed for the line. Figure 7 shows the SPICE program together with the equivalent circuit and values of the elements. Finally the results of the analysis performed with the SPICE and using DFT techniques are compared in fig.8, showing good agreement.

CONCLUSIONS

A technique to develop an efficient and universal model for lossy and dispersive transmission lines has been presented. The developed model is compatible with time domain comercial programs, such as SPICE.



Fig. 5. Universal transmissión line circuit model

ACKNOWLEDGEMENTS

This work was supported by the project TIC-0023/14 of the National Programme on Information and Communication Technologies (PRONTIC).

REFERENCES

[1] R.A. Sainati and T.J. Moravec, "Estimating High Speed Circuit Interconnect Performance", *IEEE Trans. Circuits and Systems*, Vol.36, No.4, pp.533-541, April 1989.

[2] V.K. Tripathi and A. Hill, "Equivalent Circuit Modeling of Losses and Dispersion in Single and Coupled Lines for Microwave and Millimeter-wave Integrated Circuits", *IEEE Trans. Microwave Technique*, Vol.36, No.2, pp.256-262, Feb.1988.

[3] N. Balabanian and T.A. Bickart, *Electrical Network Theory*. Malabar, F.L.: Krieger, 1985.

[4] José I. Alonso, *"Análisis en el dominio del tiempo de circuitos de microondas mediante la Transformada Z. Aplicación al estudio de efectos dispersivos en líneas de transmisión"*. M.S. Thesis. UPM. September 1989.

[5] P. Pramanick and Bhartia, "An Accurate Description of Dispersion in Microstrip". *Microwave Journal*, pp.89-96, December 1983.

[6] R.E. Collin, *"Foundations for Microwave Engineering"*, McGraw-Hill, 1966, pp.87-89.



Fig.6. Propagation characteristics of the equivalent circuit model. (———) Computed for the line; (+ + + +) Values calculated for the model



figure 7.a. Network synthesis of the attenuation function



figure 7.b. Network synthesis of the propagation function

R	Ω	53 7
Ra	Ω	4.76728655
Rb	Ω	0.87580327
La	pH	53.1985759
Lb	pH	93.0902279
Ra'	Ω	604.903954
Rb'	Ω	3292.62299
Ca'	pF	0.01844809
Cb'	pF	0.03228163

L	nH	0.35007474
C	pF	0 09221530
L1	nH	0.55310334
C1	pF	0.08538790
C2	pF	0.18443070

values of the elements

TABLE

SPICE INPUT DATA FILE FOR THE DEVELOP MODEL

```
*** GAUSSIAN PULSE ***

VIN1 5 0 PWL( 410ps .0000 415ps .0001 420ps .0002 425ps .0005 430ps .0012
+ 435ps .0031 440ps .0072 445ps .0159 450ps .0326 455ps .0625 460ps .1118
+ 465ps .1869 470ps .2916 475ps .4250 480ps .5783 485ps .7349 490ps .8720
+ 495ps .9663 500ps 1.000 505ps .9663 510ps .8720 515ps .7349 520ps .5783
+ 525ps .4250 530ps .2916 535ps .1869 540ps .1118 545ps .0625 550ps .0326
+ 555ps .0159 560ps .0072 565ps .0031 570ps .0012 575ps .0005 580ps .0002
+ 585ps .0001 590ps .0000 )

*** UNIVERSAL EQUIVALENT CIRCUIT MODEL ***

RG   5   1   53.7

RO1    1   2   53.7
RO2    7   9   53.7
RO3   10  13   53.7
RO4   12  17   53.7

EA    3   2   2   1    1
E1    4   3   0   1    1
ETB   0   4   7   9    1
E1I   7   6   1   0    2
EAI   6   0   1   2    2

EB   15   12   12   17    1
E2   16   15    0   17    1
ETA  16    0   10   13    1
E20  13   14   17    0    2
EBO  14    0   17   12    2

T1    9    0   25    0  ZO=53.7  TD=858.099165ps
X1   25    0   10    SIMUL
RL   17    0   53.7

.SUBCKT SIMUL 1 2 3
XX1 1 2 4 CIRC
XX2 4 2 5 CIRC
XX3 5 2 3 CIRC
.ENDS SIMUL
.SUBCKT CIRC 1 2 8
XXX1 1 2 3 ATEN
XXX2 3 2 4 PTI
XXX3 4 2 5 PTI
XXX4 5 2 6 PTI
XXX5 6 2 7 PTI
XXX6 7 2 8 PTI
.ENDS CIRC
.SUBCKT ATEN 1 2 3
R1 1 4 53.70
R2 4 3 53.70
RA1 1 3 4.767286
LA1 1 5 53.1985759pH
RA2 5 3 .8758032
LA2 5 3 93.0902279pH
RA3 4 6 604.903954
RA4 6 7 3.29262297K
CA1 6 2 .018448092pF
CA2 7 2 .032281530pF
.ENDS ATEN
.SUBCKT PTI 1 2 3
CPT1 1 4 0.0922154pF
CPT2 4 3 0.0922153pF
CPT3 4 5 0.0853879pF
RPT3 4 5 1000MEG
CPT4 6 2 0.1844307pF
LPT1 1 6 .35007474nH
LPT2 6 3 .35007474nH
LPT3 5 2 .55310334nH
.ENDS PTI
.PROBE v(17)
.TRAN   5.000p    2.000n    1.300n    0
.PRINT TRAN V(17)
.END
```



Fig. 8. Gaussian DC pulse dispersion at a distance L=122,142mm. (----) Response of the model developed with SPICE; (——) Computed by using DFT techniques